Pensieve header: The category GDO and the Heisenberg algebra. Based on AlexanderFromHeisenberg.nb at pensieve://Talks/LearningSeminarOnCategorification-2006/.

In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Classes\\21-1350-KnotTheory"];
<< Common.m
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

In[∘]:=
```
ser = Series[E^((a+b) x), {x, 0, 3}]
```

Out[∘]= $1+(a+b)\,x+\frac{1}{2}(a+b)^2x^2+\frac{1}{6}(a+b)^3x^3+O[x]^4$

In[∘]:=
```
Expand[ser]
```

Out[∘]= $1+(a+b)\,x+\frac{1}{2}(a+b)^2x^2+\frac{1}{6}(a+b)^3x^3+O[x]^4$

In[∘]:=
```
ser // FullForm
```

Out[∘]//FullForm=
```
SeriesData[x, 0, List[1, Plus[a, b], Times[Rational[1, 2], Power[Plus[a, b], 2]],
  Times[Rational[1, 6], Power[Plus[a, b], 3]]], 0, 4, 1]
```

A hack to fix how Expand acts on Series:

In[∘]:=
```
Unprotect[SeriesData];
SeriesData /: Expand[sd_SeriesData] := MapAt[Expand, sd, 3];
Protect[SeriesData];
```

In[∘]:=
```
Expand[ser]
```

Out[∘]= $1+(a+b)\,x+\left(\frac{a^2}{2}+a\,b+\frac{b^2}{2}\right)x^2+\left(\frac{a^3}{6}+\frac{a^2\,b}{2}+\frac{a\,b^2}{2}+\frac{b^3}{6}\right)x^3+O[x]^4$

In[∘]:=
```
{p*, x*, π*, ξ*} = {π, ξ, p, x}; (u_i_)* := (u*)_i; l_List* := #* & /@ l;
```

In[∘]:=
```
(𝔾_{A1_→B1_}[L1_] // 𝔾_{A2_→B2_}[L2_]) /; (B1* === A2) := Module[{bar, r, x, ξ},
  r = Expand[(L1 /. Table[x → bar[x], {x, B1}]) (L2 /. Table[ξ → bar[ξ], {ξ, A2}])];
  Do[r = r /. bar[x]^p_. bar[x*]^p_. → p! /. bar[x | x*] → 0, {x, B1}];
  𝔾_{A1→B2}[r]
 ]
```

In[∘]:=
```
𝔾_{{ξ_1}→{x_1}}[1 + ξ_1 x_1 + ξ_1^2 x_1^2] // 𝔾_{{ξ_1}→{x_2}}[1 + ξ_1 x_2 + ξ_1^2 x_2^2]
```

In[∘]:=
```
𝔾_{{ξ_1}→{x_2}}[1 + x_2 ξ_1 + 2 x_2^2 ξ_1^2]
```

Out[∘]= $\mathbb{G}_{\{\xi_1\}\to\{x_2\}}[1+x_2\,\xi_1+2\,x_2^2\,\xi_1^2]$

In[∘]:=
```
𝔼_s1_[ωQ1__] ≡ 𝔼_s2_[ωQ2__] := s1 === s2 ∧ Simplify[{ωQ1} == {ωQ2}]
```

In[∘]:=
```
𝔼_{A1_→B1_}[ω1_, Q1_] 𝔼_{A2_→B2_}[ω2_, Q2_] ^:= 𝔼_{A1⋃A2→B1⋃B2}[ω1 ω2, Q1 + Q2]
```

In[∘]:=
```
CF = ExpandNumerator@*ExpandDenominator@*PowerExpand@*Factor;
```

In[∘]:=
```
(𝔼_{A1_→B1_}[ω1_, Q1_] // 𝔼_{A2_→B2_}[ω2_, Q2_]) /; (B1* === A2) :=
 Module[{i, j, E1, F1, G1, E2, F2, G2, I, M = Table},
  I = IdentityMatrix@Length@B1;
  E1 = M[∂_{i,j}Q1, {i, A1}, {j, B1}]; E2 = M[∂_{i,j}Q2, {i, A2}, {j, B2}];
  F1 = M[∂_{i,j}Q1, {i, A1}, {j, A1}]; F2 = M[∂_{i,j}Q2, {i, A2}, {j, A2}];
  G1 = M[∂_{i,j}Q1, {i, B1}, {j, B1}]; G2 = M[∂_{i,j}Q2, {i, B2}, {j, B2}];
  𝔼_{A1→B2}[CF[ω1 ω2 Det[I - F2.G1]^(-1/2)], CF@Plus[
     If[A1 === {} ∨ B2 === {}, 0, A1.E1.Inverse[I - F2.G1].E2.B2],
     If[A1 === {}, 0, 1/2 A1.(F1 + E1.F2.Inverse[I - G1.F2].E1^T).A1],
     If[B2 === {}, 0, 1/2 B2.(G2 + E2^T.G1.Inverse[I - F2.G1].E2).B2]]]]
```

In[∘]:=
```
Random𝔼_{A_→B_}[r_] := Module[{ri},
  ri := RandomInteger[{-r, r}];
  𝔼_{Table[ξ_i,{i,A}]→Table[x_j,{j,B}]}[
   ri,
   Sum[ri ℏ ξ_i ξ_j, {i, A}, {j, A}] +
    Sum[ri ℏ ξ_i x_j, {i, A}, {j, B}] + Sum[ri ℏ x_i x_j, {i, B}, {j, B}]
   ]]
```

In[∘]:=
```
E1 = Random𝔼_{{1,2}→{1,2,3,4}}[5]
```

In[∘]:=
```
E2 = Random𝔼_{{1,2,3,4}→{1,2,3}}[5]
```

In[∘]:=
```
E1 // E2
```

In[∘]:=
```
𝔼Series[𝔼_{s_}[ω_, Q_], d_] := 𝔾_s[Expand@Series[ω Exp[Q], {ℏ, 0, d}]]
```

In[∘]:=
```
𝔼Series[E1, 1]
```

In[∘]:=
```
𝔼Series[E2, 1]
```

In[∘]:=
```
𝔼Series[E1, 1] // 𝔼Series[E2, 1]
```

In[∘]:=
```
𝔼Series[E1 // E2, 1]
```

In[∘]:=
```
𝔼Series[E1 // E2, 5] == (𝔼Series[E1, 5] // 𝔼Series[E2, 5])
```

In[∘]:=
```
A_ \ B_ := Complement[A, B];
(𝔼_{A1_→B1_}[ω1_, Q1_] // 𝔼_{A2_→B2_}[ω2_, Q2_]) /; (B1* =!= A2) :=
  𝔼_{A1⋃(A2\B1*)→B1⋃A2*}[ω1, Q1 + Sum[ξ* ξ, {ξ, A2 \ B1*}]] //
   𝔼_{B1*⋃A2→B2⋃(B1\A2*)}[ω2, Q2 + Sum[z* z, {z, B1 \ A2*}]]
```

A proof of the formula for R is at http://drorbn.net/cat20.

In[∘]:=
```
R_{i_,j_} := 𝔼_{{}→{p_i,x_i,p_j,x_j}}[T^{1/2}, (1 - T) p_j x_j + (T - 1) p_i x_j];
R̄_{i_,j_} := 𝔼_{{}→{p_i,x_i,p_j,x_j}}[T^{-1/2}, (1 - T^{-1}) p_j x_j + (T^{-1} - 1) p_i x_j];
C_{i_} := 𝔼_{{}→{p_i,x_i}}[T^{1/2}, 0]; C̄_{i_} := 𝔼_{{}→{p_i,x_i}}[T^{-1/2}, 0];
```

In[∘]:=
```
η_{i_} := 𝔼_{{}→{p_i,x_i}}[1, 0]
```

In[∘]:=
```
m_{i_,j_→k_} := 𝔼_{{π_i,ξ_i,π_j,ξ_j}→{p_k,x_k}}[1, -ξ_i π_j + (π_i + π_j) p_k + (ξ_i + ξ_j) x_k]
```

In[∘]:=
```
𝔼_{{}→vs_}[ωi_, Q_]_h := Module[{ps, xs, M},
   ps = Cases[vs, p_]; xs = Cases[vs, x_];
   M = Table[ωi, 1 + Length@ps, 1 + Length@xs];
   M[[2 ;;, 2 ;;]] = Table[CF[∂_{i,j}Q], {i, ps}, {j, xs}];
   M[[2 ;;, 1]] = ps; M[[1, 2 ;;]] = xs;
   MatrixForm[M]_h]
```

Reidemeister 3

In[∘]:=
```
(R_{1,2} R_{4,3} R_{5,6} // m_{1,4→1} m_{2,5→2} m_{3,6→3}) ≡ (R_{2,3} R_{1,6} R_{4,5} // m_{1,4→1} m_{2,5→2} m_{3,6→3})
```

Reidemeister 2

In[∘]:=
```
{(R̄_{1,2} R_{3,4} // m_{1,3→1} m_{2,4→2}) ≡ η_1 η_2,
 (R_{1,4} R̄_{5,2} C̄_3 // m_{2,4→2} // m_{1,3→1} // m_{1,5→1}) ≡ C̄_1 η_2}
```

Reidemeister 1’s



In[∘]:=
```
{(C̄_2 R_{1,3} // m_{1,2→1} // m_{1,3→1}) ≡ η_1, (C̄_2 R̄_{3,1} // m_{1,2→1} // m_{1,3→1}) ≡ η_1,
 (C_2 R̄_{1,3} // m_{1,2→1} // m_{1,3→1}) ≡ η_1, (C_2 R_{3,1} // m_{1,2→1} // m_{1,3→1}) ≡ η_1}
```

The “First Tangle”



In[ ]:= Factor /@ $\left(z = R_{1,6}\,\overline{C}_3\,\overline{R}_{7,4}\,\overline{R}_{5,2}\ //\ m_{1,3\to1}\ //\ m_{1,4\to1}\ //\ m_{1,5\to1}\ //\ m_{1,6\to1}\ //\ m_{2,7\to2}\right)$

In[ ]:= z_h

```
In[ ]:= ZF[Knot[4, 1]]

In[ ]:= Alexander[Knot[4, 1]][T]

In[ ]:= Timing@Table[Simplify[Alexander[K][T] ZF[K][[1]]], {K, AllKnots[{3, 9}]}]

In[ ]:= Timing@ZF[GST48]
```